



Résolution RP-3

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```

> restart;
> with(LinearAlgebra,GenerateMatrix,ReducedRowEchelonForm,
BackwardSubstitute):
with(plottools,disk):
with(plots,display):
with(algcurves,plot_real_curve):
>

```

No. 1

Soit l'équation générale du second degré à deux variables x et y :

$$ax^2 + bxy + cy^2 + dx + ey + f = 0 \quad (1)$$

Dans le cas où au moins un des paramètres a , b ou c est non nul, le tracé obtenu est appelé une conique. Dans des cas exceptionnels, le tracé est appelé conique dégénérée. Les seuls cas possibles de coniques dégénérées sont le plan cartésien au complet, l'ensemble vide, une droite, deux droites concourantes et un seul point.

Assumant que l'équation générale du second degré ne représente pas une conique dégénérée, la conique peut être identifiée de la manière suivante:

- si $b^2 - 4ac < 0$, la conique est une ellipse (Si $a = c$ et $b = 0$, cette conique est un cercle).
- si $b^2 - 4ac = 0$ la conique est une parabole
- si $b^2 - 4ac > 0$, la conique est une hyperbole

Exemples:

- Lieu_1 = $2x^2 + 8xy + 7y^2 + x - 3y + 8 = 0$
- Lieu_2 = $2x^2 + 9xy + 7y^2 + x - 3y + 8 = 0$
- Lieu_3 = $2x^2 + 7xy + 7y^2 + x - 3y + 8 = 0$

$$\text{Lieu}_1: 2x^2 + 8xy + 7y^2 + x - 3y + 8 = 0$$

```

> b^2-4*a*c=8^2-4*2*7;

```

$$-4ac + b^2 = 8$$

(2.1)


Le Lieu_1 est donc une hyperbole.

```

> Lieu_1:=2*x^2+8*x*y+7*y^2+x-3*y+8=0;
plot_real_curve(lhs(Lieu_1),x,y);

```

$$\text{Lieu}_1 := 2x^2 + 8xy + 7y^2 + x - 3y + 8 = 0$$



Nous pouvons obtenir directement la nature de ce lieu géométrique à l'aide de la macro-commande `conic` de l'extension `geometry`.

```
> geometry[conic](c1,Lieu_1,[x,y]);
geometry[form](c1);
```

c1
hyperbola2d

(2.2)

Lieu_2: $2x^2 + 8xy + 9y^2 + x - 3y + 8 = 0$

```
> b^2-4*a*c=8^2-4*2*9;
```


$$-4ac + b^2 = -8$$

(2.3)

Le *Lieu_2* est donc une ellipse.

```
> Lieu_2:=2*x^2+8*x*y+9*y^2+x-3*y+8=0;
plot_real_curve(lhs(Lieu_2),x,y,view=[-10..10,-10..10]);
```

$$\textit{Lieu_2} := 2x^2 + 8xy + 9y^2 + x - 3y + 8 = 0$$



Mais, dans ce cas-ci, il s'agit d'une ellipse dégénérée. Plus particulièrement, c'est une ellipse imaginaire et donc, il ne peut y avoir un tracé (réel) dans le plan cartésien.

```
> geometry[conic](c1,Lieu_2,[x,y]);
```

```
conic: there is no locus
```

c1

(2.4)

Lieu_3: $2x^2 + 8xy + 8y^2 + x - 3y + 8 = 0$

```
> b^2-4*a*c=8^2-4*2*8;
```

$$-4ac + b^2 = 0$$

(2.5)

Le Lieu_3 est donc une parabole.

```
> Lieu_3:=2*x^2+8*x*y+8*y^2+x-3*y+8=0;
```

```
geometry[conic](c1,Lieu_3,[x,y]):
```


```
geometry[form](c1);
```

$$\text{Lieu}_3 := 2x^2 + 8xy + 8y^2 + x - 3y + 8 = 0$$

parabola2d

(2.6)

```
> plot_real_curve(lhs(Lieu_3),x,y,view=[-10..10,-10..10]);
```



No. 2

Identifions la conique passant par les points $(-4,1)$, $(-1,2)$, $(3,2)$, $(5,1)$ et $(7,-1)$.

Obtenons le système d'équations à résoudre.

```
> Équation_générale:=a*x^2+b*x*y+c*y^2+d*x+e*y+f = 0;
```

$$\text{Équation_générale} := ax^2 + bxy + cy^2 + dx + ey + f = 0$$

(3.1)

Considérons les cinq points pris à la fois.

```
> Éq1:=eval(Équation_générale,[x=-4,y=1]);
```

```
Éq2:=eval(Équation_générale,[x=-1,y=2]);
```

```
Éq3:=eval(Équation_générale,[x=3,y=2]);
```

```
Éq4:=eval(Équation_générale,[x=5,y=1]);
```

```
Éq5:=eval(Équation_générale,[x=7,y=-1]);
```

$$\text{Éq1} := 16a - 4b + c - 4d + e + f = 0$$

$$\text{Éq2} := a - 2b + 4c - d + 2e + f = 0$$

$$\text{Éq3} := 9a + 6b + 4c + 3d + 2e + f = 0$$

$$\begin{aligned} \dot{Eq}4 &:= 25a + 5b + c + 5d + e + f = 0 \\ \dot{Eq}5 &:= 49a - 7b + c + 7d - e + f = 0 \end{aligned} \quad (3.2)$$

Créons la matrice augmentée de ce système.

```
> S:=[Éq | (1..5)];
Variables:=[a,b,c,d,e,f]:
`A|B`:= GenerateMatrix(S,Variables,augmented=true);
S:= [16a-4b+c-4d+e+f=0,a-2b+4c-d+2e+f=0,9a+6b+4c+3d+2e+f
=0,25a+5b+c+5d+e+f=0,49a-7b+c+7d-e+f=0]
```

$$A|B := \begin{bmatrix} 16 & -4 & 1 & -4 & 1 & 1 & 0 \\ 1 & -2 & 4 & -1 & 2 & 1 & 0 \\ 9 & 6 & 4 & 3 & 2 & 1 & 0 \\ 25 & 5 & 1 & 5 & 1 & 1 & 0 \\ 49 & -7 & 1 & 7 & -1 & 1 & 0 \end{bmatrix} \quad (3.3)$$

Obtenons la forme échelon-réduite de Gauss.

```
> `A|B_GJ`:=ReducedRowEchelonForm(`A|B`);
```

$$A|B_{GJ} := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & \frac{3}{113} & 0 \\ 0 & 1 & 0 & 0 & 0 & -\frac{3}{113} & 0 \\ 0 & 0 & 1 & 0 & 0 & -\frac{1}{113} & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & \frac{54}{113} & 0 \end{bmatrix} \quad (3.4)$$

Le système à résoudre est donc un système compatible ayant une infinité de solutions. Formulons-les.

```
> Racines:=BackwardSubstitute(`A|B_GJ`);
```

$$Racines := \begin{bmatrix} -\frac{3_{_t_1}}{113} \\ \frac{3_{_t_1}}{113} \\ \frac{_{t_1}}{113} \\ 0 \\ -\frac{54_{_t_1}}{113} \\ _{t_1} \end{bmatrix} \quad (3.5)$$

```
> Racines_reformulées:=subs(_t[1]=t,eval(Racines));
```

$$Racines_reformul\acute{e}es := \begin{bmatrix} -\frac{3t}{113} \\ \frac{3t}{113} \\ \frac{t}{113} \\ 0 \\ -\frac{54t}{113} \\ t \end{bmatrix} \quad (3.6)$$

On obtient $a = -\frac{3t}{113}$, $b = \frac{3t}{113}$, $c = \frac{t}{113}$, $d = 0$, $e = -\frac{54t}{113}$ et $f = t$.

```
> Solution:=zip((x,y)->x=y,Variables,convert(Racines_reformul\acute{e}es,list))
;
```

$$Solution := \left[a = -\frac{3t}{113}, b = \frac{3t}{113}, c = \frac{t}{113}, d = 0, e = -\frac{54t}{113}, f = t \right] \quad (3.7)$$

Posons $t = 113$ afin d'\acute{e}liminer les fractions, on obtient l'\acute{e}quation suivante de la conique:

```
> eval(Solution,t=113);
```

$$[a = -3, b = 3, c = 1, d = 0, e = -54, f = 113] \quad (3.8)$$

```
> assign((3.8));
> \acute{E}quation_de_la_conique:=\acute{E}quation_g\acute{e}n\acute{e}r\acute{a}le;
```

$$\acute{E}quation_de_la_conique := -3x^2 + 3xy + y^2 - 54y + 113 = 0 \quad (3.9)$$

Obtenons le signe du calcul de $b^2 - 4ac$.

```
> 'b^2-4*a*c'=b^2-4*a*c;
```


$$-4ac + b^2 = 21 \quad (3.10)$$

Nous sommes maintenant en mesure d'identifier la conique passant par ces cinq points: c'est une hyperbole car $b^2 - 4ac > 0$.

Illustrons dans un m\^eme graphique les cinq points et l'hyperbole d'\acute{e}quation

$$-3x^2 + 3xy + y^2 - 54y + 113 = 0.$$

```
> P1:=disk([-4,1], .7, color=orange):
P2:=disk([-1,2], .7, color=orange):
P3:=disk([3,2], .7, color=orange):
P4:=disk([5,1], .7, color=orange):
P5:=disk([7,-1], .7, color=orange):
> Hyperbole:=plot_real_curve(lhs(\acute{E}quation_de_la_conique),x,y,view=[-20.
.30,-20..40]):
> display({P|(1..5),Hyperbole},axesfont=[TIMES,ROMAN,8],scaling=
constrained);
```



```
> unassign('a,b,c,d,e,f,t');
```

No. 3

D'une part, puisqu'il nous faut ajuster un cercle passant par trois points, cela impose que $a = c$ et $b = 0$ dans l'équation générale $ax^2 + bxy + cy^2 + dx + ey + f = 0$. D'où

$$ax^2 + ay^2 + dx + ey + f = 0 \quad (2)$$

Déterminons l'équation du cercle passant par les points (4,3), (1,2) et (2,0).

```
> P1:=[4,3];
    P2:=[1,2];
    P3:=[2,0];
```

$$P1 := [4, 3]$$

$$P2 := [1, 2]$$

$$P3 := [2, 0]$$

(4.1)

Obtenons le système d'équations à résoudre.

```
> Équation_Générale_Cercle:=a*x^2+a*y^2+d*x+e*y+f= 0;
    Équation_Générale_Cercle := ax^2 + ay^2 + dx + ey + f = 0
```

(4.2)

Considérons les trois points pris à la fois.

```
> Éq1:=eval(Équation_Générale_Cercle,[x=op(1,P1),y=op(2,P1)]);
    Éq2:=eval(Équation_Générale_Cercle,[x=op(1,P2),y=op(2,P2)]);
    Éq3:=eval(Équation_Générale_Cercle,[x=op(1,P3),y=op(2,P3)]);
    Éq1 := 25a + 4d + 3e + f = 0
    Éq2 := 5a + d + 2e + f = 0
    Éq3 := 4a + 2d + f = 0
```

(4.3)

Créons la matrice augmentée de ce système.

```
> S:=[Éq | (1..3)];
    Variables:=[a,d,e,f];
    `A|B`:= GenerateMatrix(S,Variables,augmented=true);
    S := [25a + 4d + 3e + f = 0, 5a + d + 2e + f = 0, 4a + 2d + f = 0]
    Variables := [a, d, e, f]
```

$$A|B := \begin{bmatrix} 25 & 4 & 3 & 1 & 0 \\ 5 & 1 & 2 & 1 & 0 \\ 4 & 2 & 0 & 1 & 0 \end{bmatrix} \quad (4.4)$$

Obtenons la forme échelon-réduite de Gauss.

```
> `A|B_GJ`:=ReducedRowEchelonForm(`A|B`);
```

$$A|B_{GJ} := \begin{bmatrix} 1 & 0 & 0 & -\frac{7}{50} & 0 \\ 0 & 1 & 0 & \frac{39}{50} & 0 \\ 0 & 0 & 1 & \frac{23}{50} & 0 \end{bmatrix} \quad (4.5)$$

Le système à résoudre est donc un système compatible ayant une infinité de solutions. Formulons-les.

```
> Racines:=BackwardSubstitute(`A|B_GJ`);
```

$$Racines := \begin{bmatrix} \frac{7_tI_1}{50} \\ -\frac{39_tI_1}{50} \\ -\frac{23_tI_1}{50} \\ _tI_1 \end{bmatrix} \quad (4.6)$$

```
> Racines:=subs(op(0,op(1,indets(Racines)))=t,Racines);
```

```
Racines_reformulées:=subs(t[1]=t,Racines);
```

$$Racines := \begin{bmatrix} \frac{7t_1}{50} \\ -\frac{39t_1}{50} \\ -\frac{23t_1}{50} \\ t_1 \end{bmatrix}$$

$$Racines_reformulées := \begin{bmatrix} \frac{7t}{50} \\ -\frac{39t}{50} \\ -\frac{23t}{50} \\ t \end{bmatrix} \quad (4.7)$$

On obtient

$$a = \frac{7t}{50}, d = -\frac{39t}{50}, e = -\frac{23t}{50} \text{ et } f = t.$$

```
> Solution:=zip((x,y)->x=y,Variables,convert(Racines_reformulées,list))
;
```

$$\text{Solution} := \left[a = \frac{7t}{50}, d = -\frac{39t}{50}, e = -\frac{23t}{50}, f = t \right] \quad (4.8)$$

Quelque soit la valeur de $t \neq 0$, les coefficients a, d, e et f sont multipliés par t et on obtient alors des équations équivalentes du cercle: le cercle est donc uniquement déterminé par ces trois points.

Posons $t = 50$ afin d'éliminer les fractions l'équation de ce cercle, on obtient l'équation suivante :

```
> eval(Solution,t=50);
```


$$[a = 7, d = -39, e = -23, f = 50] \quad (4.9)$$

```
> Équation_du_cercle:=eval(Équation_Générale_Cercle,(4.9));
```

$$\text{Équation_du_cercle} := 7x^2 + 7y^2 - 39x - 23y + 50 = 0 \quad (4.10)$$

Illustrons dans un même graphique les trois points et le cercle d'équation $7x^2 + 7y^2 - 39x - 23y + 50 = 0$.

```
> P_Cercle1:=disk(P1, .1, color=orange):
P_Cercle2:=disk(P2, .1, color=orange):
P_Cercle3:=disk(P3, .1, color=orange):
> Cercle:=plot_real_curve(lhs(Équation_du_cercle),x,y,view=[-1..6,-1..4]):
> display({P_Cercle || (1..3),Cercle},axesfont=[TIMES,ROMAN,8],scaling=
constrained);
```



No. 4

Nous allons montrer qu'il existe une infinité de paraboles passant par les trois points donnés.

Obtenons le système d'équations à résoudre.

$$\begin{aligned} & \text{Équation_Générale_Conique} := a \cdot x^2 + b \cdot x \cdot y + c \cdot y^2 + d \cdot x + e \cdot y + f = 0; \\ & \text{Équation_Générale_Conique} := ax^2 + bxy + cy^2 + dx + ey + f = 0 \end{aligned} \quad (5.1)$$

Considérons à nouveau les trois points pris à la fois.

$$\begin{aligned} & P1 := [4, 3]; \\ & P2 := [1, 2]; \\ & P3 := [2, 0]; \\ & \text{Éq1} := \text{eval}(\text{Équation_Générale_Conique}, [x = \text{op}(1, P1), y = \text{op}(2, P1)]); \\ & \text{Éq2} := \text{eval}(\text{Équation_Générale_Conique}, [x = \text{op}(1, P2), y = \text{op}(2, P2)]); \\ & \text{Éq3} := \text{eval}(\text{Équation_Générale_Conique}, [x = \text{op}(1, P3), y = \text{op}(2, P3)]); \\ & \text{Éq1} := 16a + 12b + 9c + 4d + 3e + f = 0 \\ & \text{Éq2} := a + 2b + 4c + d + 2e + f = 0 \\ & \text{Éq3} := 4a + 2d + f = 0 \end{aligned} \quad (5.2)$$

Créons la matrice augmentée de ce système.

$$\begin{aligned} & S := [\text{Éq} \mid (1..3)]; \\ & \text{Variables} := [a, b, c, d, e, f]; \\ & \text{`A|B`} := \text{GenerateMatrix}(S, \text{Variables}, \text{augmented} = \text{true}); \\ & S := [16a + 12b + 9c + 4d + 3e + f = 0, a + 2b + 4c + d + 2e + f = 0, 4a + 2d + f = 0] \\ & \text{Variables} := [a, b, c, d, e, f] \\ & A|B := \begin{bmatrix} 16 & 12 & 9 & 4 & 3 & 1 & 0 \\ 1 & 2 & 4 & 1 & 2 & 1 & 0 \\ 4 & 0 & 0 & 2 & 0 & 1 & 0 \end{bmatrix} \end{aligned} \quad (5.3)$$

Obtenons la forme échelon-réduite de Gauss.

$$\begin{aligned} & \text{`A|B_GJ`} := \text{ReducedRowEchelonForm}(\text{`A|B`}); \\ & A|B_GJ := \begin{bmatrix} 1 & 0 & 0 & \frac{1}{2} & 0 & \frac{1}{4} & 0 \\ 0 & 1 & 0 & -\frac{41}{60} & -\frac{1}{5} & -\frac{5}{8} & 0 \\ 0 & 0 & 1 & \frac{7}{15} & \frac{3}{5} & \frac{1}{2} & 0 \end{bmatrix} \end{aligned} \quad (5.4)$$

Le système à résoudre est donc un système compatible ayant une infinité de solutions. Formulons-les.

$$\begin{aligned} & \text{`Racines`} := \text{BackwardSubstitute}(\text{`A|B_GJ`}); \\ & \end{aligned} \quad (5.5)$$

$$Racines := \begin{bmatrix} -\frac{t_2_3}{2} - \frac{t_2_1}{4} \\ \frac{41 t_2_3}{60} + \frac{t_2_2}{5} + \frac{5 t_2_1}{8} \\ -\frac{7 t_2_3}{15} - \frac{3 t_2_2}{5} - \frac{t_2_1}{2} \\ t_2_3 \\ t_2_2 \\ t_2_1 \end{bmatrix} \quad (5.5)$$

```
> Racines:=subs(op(0,op(1,indets(Racines)))=t,Racines);
Racines_reformulées:=convert(Variables,Vector)=subs({t[1]=t,t[2]=s,t
[3]=r},Racines);
```

$$Racines := \begin{bmatrix} -\frac{t_3}{2} - \frac{t_1}{4} \\ \frac{41 t_3}{60} + \frac{t_2}{5} + \frac{5 t_1}{8} \\ -\frac{7 t_3}{15} - \frac{3 t_2}{5} - \frac{t_1}{2} \\ t_3 \\ t_2 \\ t_1 \end{bmatrix}$$

$$Racines_reformulées := \begin{bmatrix} a \\ b \\ c \\ d \\ e \\ f \end{bmatrix} = \begin{bmatrix} -\frac{r}{2} - \frac{t}{4} \\ \frac{41 r}{60} + \frac{s}{5} + \frac{5 t}{8} \\ -\frac{7 r}{15} - \frac{3 s}{5} - \frac{t}{2} \\ r \\ s \\ t \end{bmatrix} \quad (5.6)$$

On obtient $a = -\frac{r}{2} - \frac{t}{4}$, $b = \frac{41 r}{60} + \frac{s}{5} + \frac{5 t}{8}$, $c = -\frac{7 r}{15} - \frac{3 s}{5} - \frac{t}{2}$, $d = r$, $e = s$ et $f = t$.

```
> Solution_Générale_Conique:=zip((x,y)->x=y,Variables,convert(rhs
(Racines_reformulées),list));
Solution_Générale_Conique := [a = -\frac{r}{2} - \frac{t}{4}, b = \frac{41 r}{60} + \frac{s}{5} + \frac{5 t}{8}, c = -\frac{7 r}{15} - \frac{3 s}{5} - \frac{t}{2}, d \quad (5.7)
```

$$=r, e=s, f=t$$

```
> Équation_Générale_Conique:=eval
  (Équation_Générale_Conique,`Solution_Générale_Conique`);
```

$$\text{Équation_Générale_Conique} := \left(-\frac{r}{2} - \frac{t}{4}\right)x^2 + \left(\frac{41r}{60} + \frac{s}{5} + \frac{5t}{8}\right)xy + \left(-\frac{7r}{15} - \frac{3s}{5} - \frac{t}{2}\right)y^2 + rx + sy + t = 0 \quad (5.8)$$

Déterminons les conditions sur les paramètres r, s et t afin que l'équation générale d'une conique soit celle d'une parabole.

La condition nécessaire pour que l'équation générale du deuxième degré soit celle d'une parabole (verticale, horizontale ou obliques) est que $\Delta = b^2 - 4ac = 0$.

```
> Delta:=b^2-4*a*c;
Conditions_Parabole:=eval(Delta,(5.7));
```

$$\Delta := -4ac + b^2$$

$$\text{Conditions_Parabole} := -4 \left(-\frac{r}{2} - \frac{t}{4}\right) \left(-\frac{7r}{15} - \frac{3s}{5} - \frac{t}{2}\right) + \left(\frac{41r}{60} + \frac{s}{5} + \frac{5t}{8}\right)^2 \quad (5.9)$$

Considérons le paramètre t libre dans le calcul de Δ .

```
> [solve((5.9)=0,t)];
```

$$\left[-\frac{14r}{5} - \frac{8s}{5} + \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105}, -\frac{14r}{5} - \frac{8s}{5} - \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105} \right] \quad (5.10)$$

```
> Choix_de_t:=op(1,(5.10));
```

$$\text{Choix_de_t} := -\frac{14r}{5} - \frac{8s}{5} + \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105} \quad (5.11)$$

Alors, l'équation générale d'une parabole quelconque est:

```
> local D;
Warning, A new binding for the name `D` has been created. The
global instance of this name is still accessible using the :-
prefix, :-`D`. See ?protect for details.
```

$$D \quad (5.12)$$

```
> t:=Choix_de_t;
Coefficients:=[
  A=-r/2 - t/4,
  B=(41*r)/60 + s/5 + (5*t)/8,
  C=-(7*r)/15 - (3*s)/5 - t/2,
  D=r,
  E=s,
  F=t]
```

$$\text{Coefficients} := \left[A = \frac{r}{5} + \frac{2s}{5} - \frac{4\sqrt{154r^2 + 21rs + 126s^2}}{105}, B = -\frac{16r}{15} - \frac{4s}{5} \right] \quad (5.13)$$

$$\left[\begin{aligned} & + \frac{2\sqrt{154r^2 + 21rs + 126s^2}}{21}, C = \frac{14r}{15} + \frac{s}{5} - \frac{8\sqrt{154r^2 + 21rs + 126s^2}}{105}, D = r, E = s, F \\ & = -\frac{14r}{5} - \frac{8s}{5} + \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105} \end{aligned} \right]$$

Voici donc l'équation générale de la parabole passant par les points P1,P2 et P3é

```
> Éq_Générale_Parabole:=eval(A*x^2+B*x*y+C*y^2+D*x+E*y+F=0,
Coefficients);
```

$$\begin{aligned} \text{Éq_Générale_Parabole} := & \left(\frac{r}{5} + \frac{2s}{5} - \frac{4\sqrt{154r^2 + 21rs + 126s^2}}{105} \right) x^2 + \left(-\frac{16r}{15} - \frac{4s}{5} \right. \\ & + \frac{2\sqrt{154r^2 + 21rs + 126s^2}}{21} \Big) xy + \left(\frac{14r}{15} + \frac{s}{5} - \frac{8\sqrt{154r^2 + 21rs + 126s^2}}{105} \right) y^2 + rx \\ & + sy - \frac{14r}{5} - \frac{8s}{5} + \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105} = 0 \end{aligned} \quad (5.14)$$

Simplifions l'équation générale de la parabole passant par les trois points.

```
> Éq_Générale_Parabole:=collect(simplify(Éq_Générale_Parabole),[x^2,
y^2]);
```

$$\begin{aligned} \text{Éq_Générale_Parabole} := & \left(\frac{r}{5} + \frac{2s}{5} - \frac{4\sqrt{154r^2 + 21rs + 126s^2}}{105} \right) x^2 + \left(\left(-\frac{16r}{15} - \frac{4s}{5} \right. \right. \\ & + \frac{2\sqrt{154r^2 + 21rs + 126s^2}}{21} \Big) y + r \Big) x + \left(\frac{14r}{15} + \frac{s}{5} - \frac{8\sqrt{154r^2 + 21rs + 126s^2}}{105} \right) y^2 \\ & + sy - \frac{14r}{5} - \frac{8s}{5} + \frac{16\sqrt{154r^2 + 21rs + 126s^2}}{105} = 0 \end{aligned} \quad (5.15)$$

Pour que nous ayons une parabole verticale, il faut au moins aucun terme en y^2 . Déterminons pour quelles valeurs de r le coefficient de y^2 est nul.

```
> solve({(14*r)/15 + s/5 + (8*sqrt(154*r^2 + 21*r*s + 126*s^2))/105=0},
{r});
```

$$\left\{ r = \frac{11s}{2} \right\} \quad (5.16)$$

Substituons $r = \frac{11s}{2}$ dans l'équation générale de la parabole pour éliminer le terme en y^2 .

```
> Éq_Particielière_Parabole:= subs(r = (11 *s)/2,Éq_Générale_Parabole);
```

$$\begin{aligned} \text{Éq_Particielière_Parabole} := & \left(\frac{3s}{2} - \frac{4\sqrt{4900}\sqrt{s^2}}{105} \right) x^2 + \left(\left(-\frac{20s}{3} + \frac{2\sqrt{4900}\sqrt{s^2}}{21} \right) y \right. \\ & + \frac{11s}{2} \Big) x + \left(\frac{16s}{3} - \frac{8\sqrt{4900}\sqrt{s^2}}{105} \right) y^2 + sy - 17s + \frac{16\sqrt{4900}\sqrt{s^2}}{105} = 0 \end{aligned} \quad (5.17)$$

```
> simplify(Éq_Particielière_Parabole) assuming s>0;
```


$$-\frac{s(7x^2 - 33x - 6y + 38)}{6} = 0 \quad (5.18)$$

Donc, $\forall s > 0$, la simplification nous donne une unique équation d'une parabole verticale.

```
> Parabole_V:=y=solve((5.18),y);
```

$$Parabole_V := y = \frac{7}{6}x^2 - \frac{11}{2}x + \frac{19}{3} \quad (5.19)$$

```
> Parabole_Verticale:=plots[implicitplot](Parabole_V,x=-1..6,y=-1..4):
display(Parabole_Verticale,P_Cercle|(1..3),Cercle,scaling=
constrained);
```



Similairement, pour que nous ayons une parabole horizontale, il faut au moins aucun terme en x^2 .
Déterminons pour quelles valeurs de r le coefficient de x^2 est nul

```
> solve({r/5 + (2*s)/5 - (4*sqrt(154*r^2 + 21*r*s + 126*s^2))/105},{r})
;
```

$$\left\{ r = \frac{6s}{17} \right\} \quad (5.20)$$

Substituons $r = \frac{6s}{17}$ dans l'équation générale de la parabole pour éliminer le terme en x^2 .

```
> subs(r = (6*s)/17,Éq_Générale_Parabole);
```

$$\left(\frac{8s}{17} - \frac{4\sqrt{44100}\sqrt{289}\sqrt{s^2}}{30345} \right) x^2 + \left(\left(-\frac{20s}{17} + \frac{2\sqrt{44100}\sqrt{289}\sqrt{s^2}}{6069} \right) y + \frac{6s}{17} \right) x + \left(\frac{9s}{17} - \frac{8\sqrt{44100}\sqrt{289}\sqrt{s^2}}{30345} \right) y^2 + sy - \frac{44s}{17} + \frac{16\sqrt{44100}\sqrt{289}\sqrt{s^2}}{30345} = 0 \quad (5.21)$$

```
> simplify((5.21)) assuming s>0;
```

$$\frac{s(-7y^2 + 6x + 17y - 12)}{17} = 0 \quad (5.22)$$

```
> simplify((5.21)) assuming s<0;
```

$$\frac{s(16x^2 - 40xy + 25y^2 + 6x + 17y - 76)}{17} = 0 \quad (5.23)$$

$\forall s > 0$, la simplification nous donne une unique équation d'une parabole verticale.


```
> Sol:=solve((5.22),x);
```

$$Sol := \left[\frac{7}{6}y^2 - \frac{17}{6}y + 2 \right] \quad (5.24)$$

```
> Éq_Parabole_Horizontale:=x=op(1,Sol);
```

$$Éq_Parabole_Horizontale := x = \frac{7}{6}y^2 - \frac{17}{6}y + 2 \quad (5.25)$$

```
> Parabole_Horizontale:=plots[implicitplot](Éq_Parabole_Horizontale,x=-1..6,y=-1..4):
display(Parabole_Horizontale,P_Cercle|(1..3),Cercle,scaling=constrained);
```



La parabole est oblique si le coefficient du terme en xy n'est pas nul

```
> solve(-(16*r)/15 - (4*s)/5 + (2*sqrt(154*r^2 + 21*r*s + 126*s^2))/21=0);
```

$$\left\{ r=r, s=\frac{17r}{6} \right\}, \left\{ r=\frac{11s}{2}, s=s \right\} \quad (5.26)$$

$\forall (r, s) \in \mathbb{R}^2$, telle que $s \neq \frac{17r}{6}$ ou $r \neq \frac{11s}{2}$, la parabole est oblique. Il y a donc une infinité de paraboles obliques possibles.

Par exemples.


```
> Parabole_oblique:=eval(Équation_Générale_Conique,[r=-52,s=1]);
```

$$\begin{aligned}
 \text{Parabole_oblique} := & \left(-10 - \frac{4\sqrt{415450}}{105} \right) x^2 + \left(\frac{164}{3} + \frac{2\sqrt{415450}}{21} \right) xy + \left(-\frac{145}{3} \right. \\
 & \left. - \frac{8\sqrt{415450}}{105} \right) y^2 - 52x + y + 144 + \frac{16\sqrt{415450}}{105} = 0
 \end{aligned}
 \tag{5.27}$$

```

> Parabole_Oblique_Choisie:=plots[implicitplot](Parabole_oblique,x=-1.
.5,y=-2..4):
A:=display(Parabole_Verticale,P_Cercle|(1..3)):
B:=display(Parabole_Horizontale,P_Cercle|(1..3)):
C:=display(Parabole_Oblique_Choisie,P_Cercle|(1..3)):
display(Matrix(1,3,[A,B,C]),scaling=constrained);

```



Autres solutions des numéros 3 et 4 (Mathieu Guay-Paquet et Véronique Poupart, automne 2003)

Ces deux élèves ont suivi mes cours de mathématiques avec volet enrichissement Maple. C'est sur une base volontaire que ces étudiants ont suivi mes cours. Je vous présente donc leurs solutions.

No. 3

```
> restart;with(LinearAlgebra,GenerateMatrix,ReducedRowEchelonForm,
BackwardSubstitute):
with(plottools,disk):
with(plots,display):
with(algcurves,plot_real_curve):
```

```
a)
> Équation_générale:=a*x^2 + a*y^2 + d*x + e*y + f = 0;
      Équation_générale :=  $ax^2 + ay^2 + dx + ey + f = 0$  (6.1.1)
```

```
> Points:=[[4,3],
           [1,2],
           [2,0]];
      Points := [[4,3],[1,2],[2,0]] (6.1.2)
```

```
> S:={seq( eval(Équation_générale,[x=Points[i,1],y=Points[i,2]]) ,
           i=1..3)};
      S := { $4a + 2d + f = 0$ ,  $5a + d + 2e + f = 0$ ,  $25a + 4d + 3e + f = 0$ } (6.1.3)
```

```
> Variables:=[a,d,e,f];
      Variables := [a,d,e,f] (6.1.4)
```

```
> M:=GenerateMatrix(S,Variables,augmented=true);
      M := 
$$\begin{bmatrix} 4 & 2 & 0 & 1 & 0 \\ 5 & 1 & 2 & 1 & 0 \\ 25 & 4 & 3 & 1 & 0 \end{bmatrix}$$
 (6.1.5)
```

```
> M2:=ReducedRowEchelonForm(M);
      M2 := 
$$\begin{bmatrix} 1 & 0 & 0 & -\frac{7}{50} & 0 \\ 0 & 1 & 0 & \frac{39}{50} & 0 \\ 0 & 0 & 1 & \frac{23}{50} & 0 \end{bmatrix}$$
 (6.1.6)
```

```
> Solution:=BackwardSubstitute(M2);
```

(6.1.7)

$$Solution := \begin{bmatrix} \frac{7_{-}t_1}{50} \\ -\frac{39_{-}t_1}{50} \\ -\frac{23_{-}t_1}{50} \\ -t_1 \end{bmatrix} \quad (6.1.7)$$

```
> Solution2:=zip(`=`,Variables,convert(Solution,list));
```

$$Solution2 := \left[a = \frac{7_{-}t_1}{50}, d = -\frac{39_{-}t_1}{50}, e = -\frac{23_{-}t_1}{50}, f = -t_1 \right] \quad (6.1.8)$$

```
> Solution3:=eval(Solution2,_t[1]=50);
```

$$Solution3 := [a = 7, d = -39, e = -23, f = 50] \quad (6.1.9)$$

```
> Cercle:=eval(Équation_générale,Solution3);
```

$$Cercle := 7x^2 + 7y^2 - 39x - 23y + 50 = 0 \quad (6.1.10)$$

L'équation du cercle est $7x^2 + 7y^2 - 39x - 23y + 50 = 0$.


b)

```
> Disques:=seq(disk(Points[i],.1,color=orange) , i=1..3):
```

```
> Courbe_cercle:=plot_real_curve(lhs(Cercle),x,y,view=[-1..6,-1..4])
```

```
:
```

```
> display([Disques,Courbe_cercle],scaling=constrained);
```



No. 4

a)

```
> Équation_générale:=a*x^2 + b*x*y + c*y^2 + d*x + e*y + f = 0;
```

$$\text{Équation_générale} := ax^2 + bxy + cy^2 + dx + ey + f = 0 \quad (6.2.1)$$

> 'Points'=Points;

$$\text{Points} = \begin{bmatrix} 4, 3 \\ 1, 2 \\ 2, 0 \end{bmatrix} \quad (6.2.2)$$

> S:={seq(eval(Équation_générale,[x=Points[i,1],y=Points[i,2]]) ,
i=1..3)};

$$S := \{4a + 2d + f = 0, a + 2b + 4c + d + 2e + f = 0, 16a + 12b + 9c + 4d + 3e + f = 0\} \quad (6.2.3)$$

> Variables:=[d,e,f,a,b,c];

$$\text{Variables} := [d, e, f, a, b, c] \quad (6.2.4)$$

> M:=GenerateMatrix(S,Variables,augmented=true);

$$M := \begin{bmatrix} 2 & 0 & 1 & 4 & 0 & 0 & 0 \\ 1 & 2 & 1 & 1 & 2 & 4 & 0 \\ 4 & 3 & 1 & 16 & 12 & 9 & 0 \end{bmatrix} \quad (6.2.5)$$

> M2:=ReducedRowEchelonForm(M);

$$M2 := \begin{bmatrix} 1 & 0 & 0 & \frac{33}{7} & \frac{18}{7} & \frac{6}{7} & 0 \\ 0 & 1 & 0 & \frac{6}{7} & \frac{16}{7} & \frac{17}{7} & 0 \\ 0 & 0 & 1 & -\frac{38}{7} & -\frac{36}{7} & -\frac{12}{7} & 0 \end{bmatrix} \quad (6.2.6)$$

> Solution:=BackwardSubstitute(M2);

$$\text{Solution} := \begin{bmatrix} -\frac{33_t0_3}{7} - \frac{18_t0_2}{7} - \frac{6_t0_1}{7} \\ -\frac{6_t0_3}{7} - \frac{16_t0_2}{7} - \frac{17_t0_1}{7} \\ \frac{38_t0_3}{7} + \frac{36_t0_2}{7} + \frac{12_t0_1}{7} \\ _t0_3 \\ _t0_2 \\ _t0_1 \end{bmatrix} \quad (6.2.7)$$

> Solution2:=zip(`=`,Variables,convert(Solution,list));

$$\begin{aligned} \text{Solution2} := & \left[d = -\frac{33_t0_3}{7} - \frac{18_t0_2}{7} - \frac{6_t0_1}{7}, e = -\frac{6_t0_3}{7} - \frac{16_t0_2}{7} - \frac{17_t0_1}{7}, f = \right. \\ & \left. = \frac{38_t0_3}{7} + \frac{36_t0_2}{7} + \frac{12_t0_1}{7}, a = _t0_3, b = _t0_2, c = _t0_1 \right] \end{aligned} \quad (6.2.8)$$

> Solution3:=subs([_t0[1]=t,_t0[2]=s,_t0[3]=r],Solution2);

$$\text{Solution3} := \left[d = -\frac{33r}{7} - \frac{18s}{7} - \frac{6t}{7}, e = -\frac{6r}{7} - \frac{16s}{7} - \frac{17t}{7}, f = \frac{38r}{7} + \frac{36s}{7} + \frac{12t}{7}, \right. \quad (6.2.9)$$

$$a = r, b = s, c = t$$

Pour avoir une parabole verticale, il faut avoir $a \neq 0$, $b = 0$ et $c = 0$. On a alors:

$$\begin{aligned} &> \text{Solution_verticale} := \text{eval}(\text{Solution3}, \{r=7, s=0, t=0\}); \\ &\quad \text{Solution_verticale} := [d = -33, e = -6, f = 38, a = 7, b = 0, c = 0] \end{aligned} \quad (6.2.10)$$

$$\begin{aligned} &> \text{Parabole_verticale} := \text{eval}(\text{Équation_générale}, \text{Solution_verticale}); \\ &\quad \text{Parabole_verticale} := 7x^2 - 33x - 6y + 38 = 0 \end{aligned} \quad (6.2.11)$$

Pour avoir une parabole horizontale, il faut avoir $a = 0$, $b = 0$ et $c \neq 0$. On a alors:

$$\begin{aligned} &> \text{Solution_horizontale} := \text{eval}(\text{Solution3}, \{r=0, s=0, t=7\}); \\ &\quad \text{Solution_horizontale} := [d = -6, e = -17, f = 12, a = 0, b = 0, c = 7] \end{aligned} \quad (6.2.12)$$

$$\begin{aligned} &> \text{Parabole_horizontale} := \text{eval}(\text{Équation_générale}, \text{Solution_horizontale}); \\ &\quad \text{Parabole_horizontale} := 7y^2 - 6x - 17y + 12 = 0 \end{aligned} \quad (6.2.13)$$

Les conditions pour avoir une parabole oblique sont $b \neq 0$ et $b^2 - 4ac = 0$. Comme on peut multiplier l'équation générale par une constante sans changer la courbe, on peut fixer $b = 2$.

$$\begin{aligned} &> \text{Condition} := \text{eval}(s^2 - 4*r*t = 0, s=2); \\ &\quad \text{Condition} := -4rt + 4 = 0 \end{aligned} \quad (6.2.14)$$

$$\begin{aligned} &> \text{solve}(\text{Condition}, \{r\}); \\ &\quad \left\{ r = \frac{1}{t} \right\} \end{aligned} \quad (6.2.15)$$

$$\begin{aligned} &> \text{Solution4} := \text{eval}(\text{Solution3}, \{s=2, r=1/t\}); \\ &\quad \text{Solution4} := \left[d = -\frac{33}{7t} - \frac{36}{7} - \frac{6t}{7}, e = -\frac{6}{7t} - \frac{32}{7} - \frac{17t}{7}, f = \frac{38}{7t} + \frac{72}{7} + \frac{12t}{7}, a = \frac{1}{t}, b = 2, c = t \right] \end{aligned} \quad (6.2.16)$$

$$\begin{aligned} &> \text{Parabole_oblique} := \text{eval}(\text{Équation_générale}, \text{Solution4}); \\ &\quad \text{Parabole_oblique} := \frac{x^2}{t} + 2xy + ty^2 + \left(-\frac{33}{7t} - \frac{36}{7} - \frac{6t}{7} \right)x + \left(-\frac{6}{7t} - \frac{32}{7} - \frac{17t}{7} \right)y \\ &\quad + \frac{38}{7t} + \frac{72}{7} + \frac{12t}{7} = 0 \end{aligned} \quad (6.2.17)$$

Pour chaque valeur non-nulle de t , on aura une parabole oblique différente. Il y a donc une infinité de paraboles passant par les points (4,3), (1,2) et (2,0).

b)

$$\begin{aligned} &> \text{Parabole} := \text{eval}(\text{Parabole_oblique}, t=1); \\ &\quad \text{Parabole} := x^2 + 2xy + y^2 - \frac{75}{7}x - \frac{55}{7}y + \frac{122}{7} = 0 \end{aligned} \quad (6.2.18)$$

c)

$$\begin{aligned} &> \text{Courbe_parabole} := \text{plot_real_curve}(\text{lhs}(\text{Parabole}), x, y, \text{view}=[-1..6, -1..4]); \\ &> \text{display}([\text{Disques}, \text{Courbe_cercle}, \text{Courbe_parabole}], \text{scaling}=\text{constrained}); \end{aligned}$$

